Using Maple "convex" for UNL IMMERSE 2007

Zach Teitler

6/19/07

1 Introduction

The Maple package "convex" was written by Matthias Franz. It will be useful in the UNL IMMERSE program 2007 algebra course.

2 Downloading

Download the package from the web page

`http://www-fourier.ujf-grenoble.fr/~franz/convex/`

Click on "current version." Download the file "`convex.m`".

3 Installing

It seems to be fairly non-trivial to install a Maple package.

3.1 Installing on Windows

The following instructions are intended for Windows computers, such as those found in the UNL math department network.

1. Somewhere in your home directory (`Z:` drive), create a folder called "`maplelibs`". In my case, I did this at the directory "`\\mathstat\math-zteitler\maplelibs`". I will assume that you did the same thing.
2. Move the file "`convex.m`" to that folder.
3. Open Notepad (or another similar text editor). Create a new file. Enter the following text:

```
libname:=libname,"\\\\mathstat\\math-zteitler\\maplelibs":
```

(replacing "math-zteitler" with your own username). Note that backslashes have to be escaped—that is, each backslash in the actual path to the maplelibs folder has to be written as two backslashes, in this file. If you created the `maplelibs` folder in a different location, then use that path:

```
libname:=libname,"<path-to-maplelibs-folder>":
```

(Explanation: For Maple, the `libname` variable stores the list of directories where Maple looks for library files. The command listed above adds our `maplelibs` directory to that list. Henceforth Maple will be able to find things we put in that directory.)

4. Save the file with the name "`maple.ini`" in a location where Maple will see it. For computers on the UNL math department network, it seems that the best location is the top level of your "`Z:`" drive, that is, the Samba share on the mathstat server.

3.2 Installing on Mac OS X

The following instructions record what I did on my Apple laptop. It is running OS X 10.4; I use the `tcsh` shell.

1. I created the following folder:

```
~/Library/Maple/lib/
```

2. Move the file "`convex.m`" to that folder.
3. Create a blank text file and enter the following text:

```
libname := "/Users/zteitler/Library/Maple/lib/", libname:
```

(replacing "zteitler" with your username).

4. Save the text file with the name "`.mapleinit`" in your home directory. Notice the dot at the beginning of the filename. (The file will be "invisible", that is, not visible using Finder. It can still be seen using the Terminal.)

3.3 Testing installation

To see if you have successfully installed the package, start Maple and enter the command

```
with(convex);
```

4 Documentation

Documentation for the `convex` package is available at the web site listed above. It can also be downloaded for local access. The documentation is pretty good. Since this note is intended only to help you get started using the package, you will probably have to look at the documentation eventually.

5 Using the package for IMMERSE

5.1 Loading the package

Before anything else can be done, the package has to be loaded with the command

```
with(convex);
```

5.2 Entering monomial ideals

I don't know if it's possible to directly enter a monomial ideal into Maple somehow. For now, the thing to do is to enter a polynomial with one term for each generator of the ideal.

Example 5.1. Say we want to enter the ideal $I = (x^2, xy)$. We form the polynomial $p = x^2 + xy$, and enter the following command into Maple:

```
p := x^2 + x*y
```

5.3 Producing polytopes

Given a polyonomial p representing the monomial ideal we are interested in, we can produce what Hübl's paper calls C with the command `newtonpolytope`:

```
C:=newtonpolytope(p,[x,y])
```

where `[x,y]` is a list of the variables appearing in p. (For other examples you may want to use `[x,y,z]`, or however many variables there are.)

To get what Hübl's paper calls K (which is what one might call the "Newton polyhedron"), we have to engage in some technical drudgework. The key idea is that K is just the (Minkowski) sum of C with the positive orthant of the ambient real vector space. So the following command ought to work, but doesn't:

```
K := minkowskisum(C,posorthant(2));
```

(where `2` is replaced by the dimension of the space C and K lie in, that is, the number of variables in p.) The problem is that the `convex` package makes a distinction between polytopes and cones; our C is a polytope, but `posorthant` produces a cone. To remedy this, we need a conversion.

1. Convert the positive orthant from a cone to a polyhedron:

   ```
   PO2 := convert(posorthant(2),affine);
   ```

   For 3 dimensions, change it to `posorthant(3)`. You might also want to change the name to `PO3`.

2. Make K:

   ```
   K := minkowskisum(C,PO2);
   ```

5.4 Drawing polytopes

5.5 Bounded ones

A bounded polytope can be drawn simply with the command `draw`, as in "`draw(C);`" or "`draw(C, axes=normal)`".

5.6 Unbounded ones

An unbounded polytope (or cone) can't be drawn by `draw`. The simplest way I know to draw it is to truncate the polytope by intersecting it with a reasonably-sized cube. The command "`cube(d,s)`" produces a cube of dimension d whose vertices have coordinates $(\pm s, \pm s, \ldots, \pm s)$.

Now, given an unbounded polyhedron, say K, we can make a truncation of K with the command

```
Ktrunc := intersection( cube(2,10) , K );
```

where the `2` is replaced with the dimension of K (number of variables) and the `10` is just any number that gives you a reasonable picture of K. Then to produce a drawing, use the command

```
draw(Ktrunc, axes=normal);
```

5.7 Overlays

The command `draw` can only draw one thing at a time. But it is possible to overlay two or more polytopes using the `PLOT` and `plotdata` commands.

plotdata: This command is defined by `convex.m`. It takes the same input as `draw`. Instead of actually producing a drawing, it produces all the data of a drawing. The idea is to put the `plotdata` into the argument of a `PLOT` command.

PLOT: This command makes a two-dimensional plot.

PLOT3D: This command makes a three-dimensional plot.

Example 5.2. A two-dimensional example:

```
PLOT(plotdata(C,color=black),plotdata(Ktrunc,color=gold));
```

It might also be possible to store `plotdata` in a variable. The above command puts C in the top layer and K under it, so C appears to be "on top of" K. If the order is switched, then K will cover up C and make it invisible.

Example 5.3. A three-dimensional example:

```
p3 := x^3 + x^2*y + y^2 + z^3;
C3 := newtonpolytope(p3, [x,y,z]);
PO3 := convert( posorthant(3) , affine );
K3 := minkowskisum( C3 , PO3 );
K3trunc := intersection( K3 , cube(3,10) );
PLOT3D(plotdata(C3,color=black),
    plotdata(K3trunc,[color=gold,transparency=0.5]));
```

Note that now K is drawn with transparency. In order to have more than one graphics option for a given `plotdata`, a Maple list has to be used. Such lists are marked off by brackets `[ ]`.

6 Marking the generators

You could do something like this to highlight the generators of the ideal.

```
p := x^2 + x*y ;
C := newtonpolytope(p,[x,y]) ;
gen1 := plotdata( minkowskisum( convhull([2,0]) , cube(2,1/10) ) ,
  [color=red, transparency=0.5] ) ;
gen2 := plotdata( minkowskisum( convhull([1,1]) , cube(2,1/10) ) ,
  [color=red, transparency=0.5] ) ;
PLOT(gen1,gen2,plotdata(C,[color=black,axes=normal]) ) ;
```

Explanation: We are making a little square or cube and translating it to sit at the corresponding generator. Perhaps you can find a more elegant way to do this.

7 Exercises

Exercise 7.1. Explore $C(I)$ and $K(I)$ for various I. Try to get a feeling for what they look like. Think about how you would draw them with pencil and paper to capture the same information you see on-screen.

The rest of the exercises are only SUGGESTIONS.

Exercise 7.2. Draw $C(I)$ and $K(I)$ for the following monomial ideals.

1. $I = (x, y) \subset k[x, y]$
2. $I = (x^8, x^7y^3, x^4y^4, x^3y^7, y^8)$ in $k[x, y]$
3. $I = (x^{12}, x^{11}y^3, x^{10}y^6, x^8y^4, x^7y^7, x^6y^{10}, x^4y^8, x^3y^{11}, y^{12})$ in $k[x, y]$
4. $I = (x, y) \subset k[x, y, z]$
5. $I = (x^2, y^2, z^2) \subset k[x, y, z]$
6. $I = (x^2, y^2, z^2, xyz) \subset k[x, y, z]$
7. $I = (X_0^4, X_0^3X_1^3, X_1^4) \subset k[X_0, X_1]$
8. $I = (X_0^3X_1^5, X_0^4X_1^4, X_0^5X_1^2) \subseteq k[X_0, X_1]$
9. $I = (X_0^3X_1^5, X_0^4X_1^4, X_0^5X_1^2) \subseteq k[X_0, X_1]$
10. $I = (X_0^3X_1^5, X_0^4X_1^4, X_0^5X_1^2) \subseteq k[X_0, X_1, X_2]$
11. $I = (x^4, y^3)$ in $k[x, y]$; in $k[x, y, z]$
12. $I = (x^4, y^{11}, x^2y^6)$ in $k[x, y]$; in $k[x, y, z]$

Make 'em fancy; make 'em schmancy.

Exercise 7.3. Develop a way to have Maple find $\mathfrak{m}I$. [HINT: Try multiplying $p = p_I$ by $x + y$ or $x + y + z$.]

Exercise 7.4. Let $I = (x^4, xy, y^5) \subset k[x, y]$. How many monomial ideals J are there with $C(J) = C(I)$? How many are there with $K(J) = K(I)$?

Exercise 7.5. Prove/find examples of the following.

a. Let $I \subset k[x, y]$ be a monomial ideal. Show that $K(I)$ has exactly two unbounded facets. Show [with examples] that the number of bounded facets can be arbitrarily high.

b. Let $I \subset k[x, y, z]$ be a monomial ideal. Show the number of unbounded facets of $K(I)$ is at least 3. Show [with examples] that the number of unbounded facets of $K(I)$ can be arbitrarily high. Find an example where $K(I)$ has exactly 5 bounded facets. Can you find an example where $K(I)$ has exactly 5 bounded facets and exactly 3 unbounded facets?

c. Generalize the statement about arbitrarily many unbounded facets of $K(I)$ from monomial ideals in $k[x, y, z]$ to monomial ideals in $k[x_1, \ldots, x_n]$ with $n \geq 3$.

d. (Harder): Given integers $a \geq 3$ and $b \geq 0$, is there an I such that $K(I)$ has exactly a unbounded facets and b bounded facets? Which pairs (a, b) can occur? What about counting unbounded/bounded edges, instead of facets?